unittest
{
	assert( [ "3 1", "1 2", "1 3", "2 3" ].parse.expand.solve == "0\n1\n2" );
	assert( [ "6 3", "2 1", "2 3", "4 1", "4 2", "6 1", "2 6", "4 6", "6 5" ].parse.expand.solve == "6\n4\n2\n0\n6\n2" );
}



import std.container;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nm = input.front.split.to!( long[] );
	auto abs = new long[ 2 ][ nm[ 0 ] - 1 + nm[ 1 ] ];
	foreach( ref ab; abs )
	{
		input.popFront;
		ab = input.front.split.to!( long[] );
		ab[] -= 1;
	}
	return tuple( nm[ 0 ], abs );
}

auto solve( long n, long[ 2 ][] abs )
{
	// 辺配列から隣接リストを作成、頂点ごとに親の数をカウント
	auto nss = new long[][ n ];
	auto ps = new long[ n ];
	foreach( ab; abs )
	{
		nss[ ab[ 0 ] ] ~= ab[ 1 ];
		ps[ ab[ 1 ] ]++;
	}
	
	// 根を検索
	auto queue = DList!long();
	foreach( i, p; ps )
	{
		if( p == 0 )
		{
			queue.insertBack( i );
			break;
		}
	}
	
	// 幅優先探索で最長経路を探す
	auto anss = new long[ n ];
	while( !queue.empty )
	{
		auto f = queue.front;
		queue.removeFront();
		
		foreach( t; nss[ f ] )
		{
			ps[ t ]--;
			if( 0 < ps[ t ] ) continue;		// 親の探索が全部終わるまで保留
			anss[ t ] = f + 1;		// 最後に到達した辺を採用
			queue.insertBack( t );
		}
	}
	return anss.to!( string[] ).join( "\n" );
}
